CHAPTER 4 Plot Control Commands

One of DATAPLOT's strengths is the ability to control the various attributes of a plot. Commands in this category specify details of subsequent plots such as line type, labels, and log scale. Examples include LINES, LABEL, and LOG. A chart at the end of this introduction highlights the most commonly used plot attributes.

The plot control commands do not have immediate effect. Instead, they set the value of internal switches. When DATAPLOT generates a subsequent plot or diagrammatic graphic, it uses the current value of these internal switches. This means that the order that the plot control commands are entered in does not matter (as long as they are entered before the desired plot). If a plot command does have to be entered before or after a specific command, this will be explicitly stated in the documentation for that command. It also means that the plot control command stays in effect for all subsequent plots until it is explicitly changed or reset to its default value. For example, entering the command TITLE SAMPLE PLOT means that SAMPLE PLOT will be the title on all subsequent plots until a new title is entered. Entering a plot control command without any arguments is one way of resetting the default value.

DATAPLOT defines a 0 to 100 coordinate system (i.e., a percentage coordinate system) in both the horizontal and vertical directions. The point (0,0) is the lower left corner of the device and (100,100) is the upper right corner. This will be referred to as the DATAPLOT coordinate system or DATAPLOT 0 to 100 coordinates. Many plot control commands specify arguments in terms of this DATAPLOT coordinate system. The user coordinate system refers to the horizontal and vertical coordinates of the most recent plot.

The WINDOW CORNER COORDINATES, MULTIPLOT, and ORIENTATION commands specify sub-areas of the device and define a new 0 to 100 coordinate system relative to this sub-area. If a sub-area is in effect, any attribute commands specified in the 0 to 100 units apply to this sub-area, not the full device area.

Most of the commands in this chapter specify an attribute for a line, text, or a region. A region refers to the interior of some type of geometric figure (e.g., the interior of a bar). In order to avoid excessive duplication, the available choices for the attribute are not given for each individual command. The next few sections specify the type of argument expected for the given attribute and gives a reference to where all the available choices are listed.

Line attributes

The line attributes that DATAPLOT supports are style (e.g., solid, dash, dotted), color, and thickness.

The available line styles are given in the Graphics Attributes chapter. Line styles are given as strings (e.g., LINE DASHED). The available colors are given in the Color chapter. Colors can be given as either a string (e.g., BLUE) or as an integer index number or parameter (e.g., LINE COLOR 3 or LINE COLOR ICOLOR). Line thickness is also discussed in the Graphics Attributes chapter. A thickness is specified as a number or parameter (e.g., LINE THICKNESS 0.3 or LINE THICKNESS AWIDTH) where the number given is in terms of the vertical DATAPLOT 0 to 100 coordinates. That is, a thickness of 0.3 means 0.3 percent of the vertical size of the given graphics device. If line thickness is available in the graphics device hardware, the requested thickness is mapped to the closest hardware thickness available on the device. If thickness is not supported in the graphics device hardware, thick lines are generated by drawing a series of parallel lines.

Text attributes

Text can either be drawn in hardware (referred to as hardware characters) or drawn as a series of stroked lines (referred to as software characters). The text attributes that DATAPLOT supports are angle or direction, case (i.e., upper, lower, or as entered), color, font, fill, justification, size (height and width), and thickness.

Software characters can be drawn at any arbitrary angle. The angle attribute, which only applies if a software character is drawn, is specified as a real number or a parameter (e.g., LEGEND ANGLE 45 or LEGEND ANGLE AX). The direction attribute, which only applies if a hardware character is drawn, can be either horizontal or vertical and is specified as a character string (e.g., LEGEND DIRECTION VERTICAL or LEGEND DIRECTION HORIZONTAL). If the graphics device can rotate character strings (e.g.,Postscript or a penplotter device), then DATAPLOT draws vertical strings by rotating the string 90 degrees counterclockwise. If the device cannot rotate characters (e.g., a Tektronix terminal), each character is plotted individually in a horizontal direction with one character per line down the page.

The case attribute specifies the default case and is entered as either UPPER, LOWER, or ASIS (e.g., LEGEND CASE ASIS). UPPER means that all characters are converted to upper case, LOWER means that all characters are converted to lower case, and ASIS means that the case is preserved as entered on the command line. DATAPLOT also supports in-line case shifting (see the Text Attributes chapter for details). In-line case shifts override the case attribute.

The available colors are given in the Color chapter. Colors can be given as either a string (e.g., BLUE) or as an index number or a parameter (e.g., LEGEND COLOR 3 or LEGEND COLOR ICOLOR).

DATAPLOT supports seven software fonts and one hardware font. The available fonts, and sample output, are documented under the FONT command in the Diagrammatic Graphics chapter. Fonts are entered as a string (e.g., TITLE FONT DUPLEX). A few devices, such as Postscript and X11, have a range of hardware fonts available. The method for specifying an alternate hardware font is device dependent and is documented in the Output Devices chapter. For example, the method for specifying a Postscript font is documented under the POSTSCRIPT command in the Output Devices chapter. For the purposes of this chapter, these hardware fonts are simply hardware characters.

DATAPLOT allows certain special symbols, e.g., circles and squares, to be entered in-line in text strings (e.g., LEGEND 1 CIRC() FIRST LINE). A few of these symbols can be solid filled. The list of the available symbols which can be filled is documented under the FILL command in the Diagrammatic Graphics chapter. The fill attribute specifies whether these special symbols are to be filled or not and are entered as an ON/OFF switch (e.g., LEGEND FILL ON). The fill attribute only applies to software characters.

The available justifications are documented under the JUSTIFICATION command in the Diagrammatic Graphics chapter. The justification attribute is entered as a string (e.g., LEGEND JUSTIFICATION LECE). The text string can be justified in both the horizontal and vertical directions.

The size is the height of the character measured from the visible bottom of the character to the visible top of the character. The vertical spacing between characters is not counted. Character sizes are specified in the DATAPLOT 0 to 100 coordinate units. That is, the character height is specified as a percentage of the vertical size of the device and the character width is specified as a percentage of the horizontal size of the device. A height of 0 would be negligibly small while a height of 100 would be full screen vertical distance. The SIZE commands only specify the height, in which case the width is automatically set to 1/2 of the height. The HW commands specify the height and width both. Character sizes are specified as numbers or parameters (e.g., TITLE SIZE 2.6 or CHARACTER HW ATEMP1 ATEMP2). Character sizes apply to both hardware and software characters. However, on a given device, hardware characters may be restricted to a few discrete sizes. In this case, DATAPLOT maps the requested size to the closest hardware size available on that device.

Software characters are drawn as lines. This means that they can be drawn with the same attributes as given above for other lines. This includes style, color, and thickness. The color is still determined from the color setting attribute command. At this time, there are no commands for setting the line style for software characters (they are always drawn with solid lines). However, there are commands for setting the line

thickness for text strings. The comments given above for setting the thickness for lines applies to the thickness attributes for characters. The thickness attribute does not apply to hardware characters.

Region attributes

Geometric figures, such as the bar on a bar chart, distinguish between the attributes for the border and for the interior. In addition, the interior can be either solid filled or filled with a cross-hatch pattern. The border and the lines used to draw the cross-hatch fills are simply lines, so setting their attributes (style, color, thickness) is exactly as described above for lines in general.

The interior of a figure has a fill attribute, which is specified as ON or OFF (e.g., BAR FILL ON). ON means the region is either solid filled or filled with some type of cross-hatch pattern while OFF means that the region is empty. The type of fill is defined by the pattern attribute which is entered as a string (e.g., BAR PATTERN HORI or BAR PATTERN D1D2). The available cross-hatch patterns are given under the REGION PATTERN command in this chapter. The spacing attribute is the distance between lines in the cross-hatch pattern. This attribute is given as a number or parameter in the 0 to 100 vertical units (e.g., BAR PATTERN SPACING 3 or BAR PATTERN SPACING ATEMP).

The only attribute for solid filled regions (other than being ON or OFF) is the color. The color for regions is specified in an analogous manner to the color for lines and text.

Solid fills for regions can either be generated in software or hardware, depending on the device. Although hardware region fills are considerably faster, they are not available or may be prone to unpredictable behavior on some devices. The Output Devices chapter documents for each specific device available in DATAPLOT whether it performs solid fills in hardware or software.

The commands in the plot control category are:

Page control

MULTIPLOT CORNER COORDINATES	Specify the location of the multi-plot region.
MULTIPLOT	Specify the number of plot regions on a page.
WINDOW CORNER COORDINATES	Specify the portion of the device area to use.
ORIENTATION	Specify whether subsequent plots are generated in landscape, portrait, or poster orientation.

Title attributes

TITLE	Specify the title to appear at the top of a plot.
TITLE AUTOMATIC	Specify an automatically generated title.
TITLE CASE	Specify the case (upper or lower) for the title.
TITLE COLOR	Specify the color for the title.
TITLE DISPLACEMEMNT	Specify the distance from the frame to the title.
TITLE FONT	Specify the font for the title.
TITLE SIZE	Specify the title size (height).
TITLE THICKNESS	Specify the thickness (when using a software font) for the title.

Axis label attributes

...LABEL	Specify axis labels to appear at the sides and the bottom of the plot.
...LABEL AUTOMATIC	Specify automatically generated labels.
...LABEL CASE	Specify the case (upper or lower) for axis labels.
...LABEL COLOR	Specify the colors for axis labels.
...LABEL DISPLACEMENT	Specify the distance from the frame to the axis labels.
...LABEL FONT	Specify the font for axis labels.
...LABEL FILL	Specify the fill switch for axis labels.
...LABEL SIZE	Specify the size (height) for axis labels.

...LABEL THICKNESS	Specify the thickness (when using a software font) for axis labels.
Legend attributes	
LEGEND ...	Specify the text for plot legends.
LEGEND ... ANGLE	Specify the angle at which legends are drawn.
LEGEND ... CASE	Specify the case (upper or lower) for legends.
LEGEND ... COLOR	Specify the color for legends.
LEGEND ... COORDINATES	Specify the position for legends.
LEGEND ... DIRECTION	Specify the direction (horizontal or vertical) for legends.
LEGEND ... FILL	Specify the fill switch for legends.
LEGEND ... FONT	Specify the font for legends.
LEGEND ... HW	Specify the size (height and width) for legends.
LEGEND ... JUSTIFICATION	Specify the justification for legends.
LEGEND ... SIZE	Specify the size (height) for legends.
LEGEND ... THICKNESS	Specify the thickness (when using a software font) for legends.
Character attributes	
CHARACTERS	Specify the plot character types (X, SQUARE, etc.).
CHARACTER ANGLE	Specify the angle at which plot characters are drawn.
CHARACTER AUTOMATIC	Specify a variable to be used as the arguments to the CHARACTER command.
CHARACTER CASE	Specify the case (upper or lower) for plot characters.
CHARACTER COLORS	Specify the colors for plot characters.
CHARACTER FILL	Specify the fill switch (on or off) for plot characters (only applies to certain character types).
CHARACTER FONT	Specify the font for plot characters.
CHARACTER HW	Specify the plot character size (height and width).
CHARACTER JUSTIFICATION	Specify the justification for plot characters.
CHARACTER MAPPING	Specify whether the index into CHARACTER command is based on the rank or the exact value of the tag variable.
CHARACTER OFFSET	Specify the offset (displacement) for plot characters.
CHARACTER SIZES	Specify the size (height) for plot characters.
CHARACTER THICKNESS	Specify the thickness (when using a software font) for plot characters.
CHARACTER WIDTH	Specify the size (width) for plot characters.
Line attributes	
LINES	Specify the plot line types (SOLID, DOT, etc.).
LINE THICKNESS	Specify the thicknesses of plot lines.
LINE COLORS	Specify the colors for plot lines.
Spike attributes	
SPIKE	Specify the existence (ON/OFF) of spikes on plots.
SPIKE BASE	Specify the base location for plot spikes.
SPIKE COLOR	Specify the color for plot spikes.
SPIKE DIRECTION	Specify the direction (horizontal or vertical) for plot spikes.
SPIKE LINE	Specify the line types for plot spikes.

SPIKE THICKNESS	Specify the thickness for plot spikes.
Bar attributes	
BAR	Specify the existence (ON/OFF) of bars on plots.
BAR BASE	Specify the base location for plot bars.
BAR BORDER COLOR	Specify the plot bar border colors.
BAR BORDER LINE	Specify the plot bar border line types.
BAR BORDER THICKNESS	Specify the plot bar border thicknesses.
BAR DIMENSION	Specify the plot bar dimensions to be 2d or 3d.
BAR DIRECTION	Specify the plot bar directions (horizontal or vertical).
BAR FILL	Specify the existence (ON/OFF) of bar fills.
BAR FILL COLOR	Specify the plot bar fill (background) colors.
BAR PATTERN	Specify the plot bar fill pattern types.
BAR PATTERN COLOR	Specify the plot bar fill pattern colors.
BAR PATTERN LINE TYPE	Specify the plot bar fill line types.
BAR PATTERN SPACING	Specify the plot bar fill line spacings.
BAR PATTERN THICKNESS	Specify the plot bar fill line thicknesses.
BAR WIDTH	Specify the widths for plot bars.
Region attributes	
REGION BASE	Specify the base locations for plot regions.
REGION FILL	Specify the existence (ON/OFF) of regions on plots.
REGION FILL COLOR	Specify the region solid fill colors.
REGION PATTERN	Specify the region fill pattern types.
REGION PATTERN COLOR	Specify the region hatch fill pattern colors.
REGION PATTERN LINE TYPE	Specify the region fill pattern line types.
REGION PATTERN SPACING	Specify the region fill pattern line spacings.
REGION PATTERN THICKNESS	Specify the region fill pattern line thicknesses.
Background attributes	
BACKGROUND COLOR	Specify the background color (inside the frame).
MARGIN COLOR	Specify the background color (outside the frame).
Frame attributes	
...FRAME	Specify the existence (ON/OFF) of the plot frame.
FRAME CORNER COORDINATES	Specify the location and shape of the plot frame.
...FRAME COLOR	Specify the colors for the plot frame.
...FRAME THICKNESS	Specify the thicknesses for the plot frame.
...FRAME PATTERN	Specify the line types for the plot frame.
Scale attributes	
...MINIMUM	Specify the minima to appear on the plot frame.
...MAXIMUM	Specify the maxima to appear on the plot frame.
...LIMITS	Specify the limits (minimum and maximum) for the plot frame.
...LOG	Specify the existence (ON/OFF) of a logarithmic scale.
...WEIBULL SCALE	Specify the existence (ON/OFF) of a Weibull scale.
Grid attributes	
...GRID	Specify the existence (ON/OFF) of grid lines.
...GRID LINE	Specify the line types of plot grid lines.

...GRID COLOR	Specify the colors of plot grid lines.
...GRID THICKNESS	Specify the thicknesses of plot grid lines.
GMINOR	Specify the existence (ON/OFF) of minor grid lines.
Tic mark attributes	
...TIC MARK	Specify the existence (ON/OFF) of tic marks.
...TIC MARK COLOR	Specify the plot tic mark colors.
...TIC MARK OFFSET	Specify the distance from the frame corner to the first and last tic marks.
TIC OFFSET UNITS	Specify the units for tic offsets (data units or DATAPLOT 0.0 to 100.0 units).
...TIC MARK POSITION	Specify the plot tic mark position (inside/outside/thru).
...TIC MARK SIZE	Specify the plot tic mark size.
...TIC MARK THICKNESS	Specify the plot tic mark thicknesses.
...MAJOR TIC MARK NUMBER	Specify the number of major tic marks.
...MINOR TIC MARK NUMBER	Specify the number of minor tic marks.
Tic mark label attributes	
...TIC MARK LABEL	Specify the existence (ON/OFF) of tic mark labels.
...TIC MARK LABEL ANGLE	Specify the plot tic mark label angles
...TIC MARK LABEL CASE	Specify the plot tic mark cases (upper or lower).
...TIC MARK LABEL COLOR	Specify the plot tic mark label colors.
...TIC MARK LABEL CONTENT	Specify alphanumeric tic mark labels.
...TIC MARK LABEL DECIMAL	Specify the number of digits to the right of the decimal point for tic mark labels.
...TIC MARK LABEL DIRECTION	Specify the plot tic mark label directions (horizontal or vertical).
...TIC MARK LABEL DISPLACEMENT	Specify the distance from the frame to tic mark labels.
...TIC MARK LABEL FONT	Specify the plot tic mark labels fonts.
...TIC MARK LABEL FORMAT	Specify the type of plot tic mark labels (real/exponential/ power/alpha).
...TIC MARK LABEL HW	Specify the plot tic mark label heights and widths.
...TIC MARK LABEL JUSTIFICATION	Specify the plot tic mark label justifications.
...TIC MARK LABEL SIZE	Specify the plot tic mark label heights.
...TIC MARK LABEL THICKNESS	Specify the plot tic mark label thicknesses.
Arrow attributes	
ARROW ... COORDINATES	Specify the location of plot arrows.
ARROW ... COLOR	Specify the colors for arrows.
ARROW ... PATTERN	Specify the line type for arrows.
ARROW ... THICKNESS	Specify the line thickness for arrows.
Box attributes	
BOX ... CORNER COORDINATES	Specify the location of plot boxes.
BOX ... COLOR	Specify the frame colors for boxes.
BOX ... PATTERN	Specify the frame line types for boxes.
BOX ... THICKNESS	Specify the frame line thicknesses for boxes.
BOX ... FILL COLOR	Specify the pattern fill colors for boxes.
BOX ... FILL GAP	Specify the pattern fill line spacings for boxes.
BOX ... FILL LINE	Specify the pattern line types for boxes.

BOX ... FILL PATTERN	Specify the pattern fill types for boxes.
BOX ... FILL THICKNESS	Specify the pattern fill line thicknesses for boxes.
BOX ... SHADOW HW	Specify the shadow sizes for boxes.
Segment attributes	
SEGMENT ... COORDINATES	Specify the location of plot line segments.
SEGMENT ... COLOR	Specify the colors for plot line segments.
SEGMENT ... PATTERN	Specify the line type for plot line segments.
SEGMENT ... THICKNEESS	Specify the line thickness for plot line segments.
3D attributes	
ROTATE EYE	Rotate the eye coordinates (EYE COORDINATES is documented in Support Commands chapter).
3DFRAME	Specify the type of frame to draw on a 3d plot.
ORIGIN COORDINATES	Specify the reference origin for a 3d plot.
PEDESTAL [not working]	Specify the existence (ON/OFF) of a pedestal on a 3d-plot.
PEDESTAL SIZE [not working]	Specify the pedestal size for a 3d plot.
PEDESTAL COLOR [not working]	Specify the pedestal color for a 3d plot.
VISIBLE [not working]	Specify whether background lines are visible on 3d plots (ON/OFF).
Design of Experiment plot attributes	
DEX DEPTH	Specify the depth of DEX interaction terms.
DEX HORIZONTAL AXIS	Specify the horizontal axis for DEX plots.
DEX WIDTH	Specify the width of levels for DEX plots.
Miscellaneous attributes	
PRE-ERASE	Specify whether subsequent plots perform an initial screen erase (ON/OFF).
BELL	Specify whether subsequent plots ring the bell before plotting (ON/OFF).
SEQUENCE	Specify whether subsequent plots contain an automatic sequence number (ON/OFF).
HARDCOPY	Specify whether subsequent plots have automatic hardcopy generated (ON/OFF).
PRE-SORT	Specify whether subsequent plots pre-sort the data before plotting (ON/OFF).
HORIZONTAL SWITCH	Specify whether plots are generated horizontally or vertically.

The ... in some of the commands indicates user-defined options for the command, as in

X1LABEL, X2LABEL, X3LABEL, Y1LABEL, Y2LABEL
LEGEND 1 COORDINATES, LEGEND 2 COORDINATES, etc.
XLOG, YLOG, X1LOG, X2LOG, Y1LOG, Y2LOG

The following program generates a chart that demonstrates the most commonly set plot attributes.

```
FEEDBACK OFF
LET Y = DATA 6.3 5.6 4.5 3.6 2.7 1.5 0.6
LET X = DATA 1 2 3 4 5 6 7
LET TAG = DATA 1 1 2 3 4 5 6
.
```

```
FRAME CORNER COOR 15 20 65 90
YLIM 0 8; XLIM 0 10
FONT SIMPLEX
.
CHARACTER CIRCLE CIRCLE BLANK BLANK BLANK USA
CHAR FILL OFF ON OFF OFF OFF OFF
CHAR HW 2 1.3 ALL
CHARACTER ANGLE 0 0 0 0 0 45
CHARACTER OFFSET 0 0 0 0 0 0 0 0 0 0 4 5
CHARACTER JUSTIFICATION CECE CECE CECE CECE CECE LEFT
LINES DOTTED BLANK BLANK BLANK BLANK BLANK
SPIKE OFF ON OFF OFF OFF OFF
SPIKE BASE 0 1 0 0 0 0
BAR BASE 0 0 1 0 0 0
BAR OFF OFF ON ON ON ON
BAR FILL OFF OFF ON ON ON ONTS
BAR PATTERN NONE NONE NONE SOLID DDDU SOLID
BAR DIMENSION 2 2 2 2 2 3
BAR WIDTH .5 ALL
X1TIC DECIMALS 1
.
TITLE PRODUCTIVITY ANALYSIS
Y1LABEL THROUGHPUT
XLABEL COMPONENT; X2LABEL NATIONAL SITES; X3LABEL JUNE TO AUGUST
LABEL SIZE 2
.
LEGEND 1 COMMERCIAL
LEGEND 2 GOVERNMENT
LEGEND 3 ACADEMIA
LEGEND 1 COOR 17 87
LEGEND 2 COOR 17 84
LEGEND 3 COOR 17 81
.
PLOT Y X TAG
.
HW 1 .5
VERTICAL SPACING 1.2
FONT TRIPLEX ITALIC
MARGIN 75
MOVE 75 89
.
TEXT TITLE
TEXT LEGEND 1; TEXT LEGEND 2; TEXT LEGEND 3
TEXT CHARACTERS; TEXT LINES
TEXT Y1TIC LABELS; TEXT Y1TICS
TEXT Y1LABEL
TEXT CHARACTER FILL
TEXT SPIKE; TEXT BAR
TEXT BAR WIDTH
TEXT BAR FILL
TEXT SPIKE BASE
TEXT BAR BASE
TEXT BAR PATTERN
TEXT CHARACTER
TEXT CHARACTER OFFSET
TEXT CHARACTER ANGLE
```

```
TEXT BAR DIMENSION
TEXT BAR FILL
TEXT; TEXT; TEXT; TEXT; TEXT; TEXT; TEXT; TEXT; TEXT; TEXT
TEXT X1TICS
TEXT X1TIC LABELS
TEXT X1TIC LABEL DECIMALS
TEXT X1MAXIMUM
TEXT X1LABEL (= XLABEL)
TEXT X2LABEL
TEXT X3LABEL
.
LINES SOLID
HW .3 1.5
LET NUMTEXT = 29
LET X1 = 74.5 FOR I = 1 1 NUMTEXT
LET Y1 = SEQUENCE 89.5 -2.2 40
LET Y1 = 18.9 -2.2 5 FOR I = 23 1 NUMTEXT
READ X2 Y2
54.9 98.0
31.8 88.1
31.8 85.0
28.8 81.9
21.7 75.2
22.9 72.6
12.4 64.3
15.2 64.0
5.0 57.7
31.4 59.9
30.7 55.3
36.8 51.6
34.8 45.8
41.9 40.0
30.7 29.4
36.8 30.0
46.9 30.2
56.3 32.2
56.3 32.2
56.3 32.2
52.7 25.9
52.7 23.8
65.4 18.3
67.9 16.3
67.9 15.4
66.0 14.2
46.7 10.8
50.4 7.7
50.4 4.7
END OF DATA
LOOP FOR K = 1 1 NUMTEXT
        LET X1P=X1(K); LET Y1P=Y1(K)
        LET X2P=X2(K); LET Y2P=Y2(K)
        ARROW X1P Y1P X2P Y2P
END OF LOOP
```


PRODUCTIVITY ANALYSIS
TITLE
COMMERCIAL
GOVERNMENT
ACADEMIA
LEGEND 1
LEGEND 2
LEGEND 3
CHARACTERS
LINES
Y1TIC LABELS
Y1TICS
Y1LABEL
CHARACTER FILL
SPIKE
BAR
BAR WIDTH
BAR FILL
SPIKE BASE
BAR BASE
BAR PATTERN
CHARACTER
CHARACTER OFFSET
CHARACTER ANGLE
BAR DIMENSION
BAR FILL
THROUGHPUT
0
1
2
3
4
5
6
7
8
USA
0.0 1.0 2.0 3.0 4.0 5.0 6.0 7.0 8.0 9.0 10.0
X1TICS
X1TIC LABELS
X1TIC LABEL DECIMALS
X1MAXIMUM
COMPONENT
NATIONAL SITES
JUNE TO AUGUST
X1LABEL (= XLABEL)
X2LABEL
X3LABEL